HEWLETT-PACKARD

Step-by-Step Solutions
For Your HP Calculator

Mechanical Engineering



HP-42S

HEWLETT PACKARD

Help Us Help You!

Please take a moment to complete this postage-paid card, tear it out and put it in the mail. Your responses and comments will help us better understand your needs and provide you with the best procedures to solve your problems. Thank you!

HELP US HELP YOU!

Book: **Mechanical Engineering** Date acquired: ____________________

Name ____

Street ____

City, State, Zip ____

Phone (_____) ____________________ Business ☐ or Home ☐

1. What calculator will you use this book with?

   009 ☐ **HP-42S** 006 ☐ **Other** ____________________

2. How many other HP solution books have you bought for this calculator? ____

3. What is your occupation?

   101 ☐ **Student** 103 ☐ **Professional** 109 ☐ **Other** ____________________

4. Where did you purchase this book?

   403 ☐ **Bookstore** 404 ☐ **Discount or Catalog Store**

   407 ☐ **Mail Order** 410 ☐ **HP Direct** 411 ☐ **Other** ____________________

5. How did you first hear about this book?

   501 ☐ **HP Owner** 503 ☐ **Advertising** 506 ☐ **HP Salesperson**

   507 ☐ **Brochure** 508 ☐ **Other** ____________________

6. To what degree did this book influence your calculator purchase decision?

   601 ☐ **Major Influence** 602 ☐ **Minor Influence** 603 ☐ **No Influence**

7. How well does this book cover the material you expected?

   701 ☐ **Good** 702 ☐ **Moderate** 703 ☐ **Low**

8. What level of knowledge is required to make use of the topics in this book?

   801 ☐ **High** 802 ☐ **Medium** 803 ☐ **Low**

9. How clearly was the material in this book presented?

   901 ☐ **Good** 902 ☐ **Moderate** 903 ☐ **Low**

10. How would you rate the value of this book for your money?

    111 ☐ **High** 112 ☐ **Medium** 113 ☐ **Low**

Comments: (Please comment on improvements and additional applications or subjects you would like HP to cover in this or another solution book.) __________

____

____

____

NO POSTAGE
NECESSARY
IF MAILED
IN THE
UNITED STATES

BUSINESS REPLY MAIL

FIRST CLASS MAIL PERMIT NO. 40 CORVALLIS, OR

POSTAGE WILL BE PAID BY ADDRESSEE

HEWLETT-PACKARD COMPANY
CALCULATOR MARKET RESEARCH
1000 NE CIRCLE BLVD.
CORVALLIS OR 97330-9988

Mechanical Engineering

Step-by-Step Solutions for Your HP-42S Calculator

HEWLETT PACKARD

Edition 1 October 1988
Reorder Number 00042-90022

Notice

Corvallis Division
1000 N.E. Circle Blvd.
Corvallis, OR 97330, U.S.A.

Printing History

Edition 1 October 1988 Mfg. No. 00042-90024

Contents

How To Use This Book

Please take a moment to familiarize yourself with the formats used in this book.

Organization. Each chapter in this book covers a different area of mechanical engineering. Sections within each chapter highlight the use of each program. These sections are organized like this:

- Description of the program, including equations and variables used.
- Special remarks and limitations.
- General instructions.
- Keystroke examples.
- Program listings.

About the Examples. Unless otherwise stated, the keystrokes and displays shown in each section assume the following conditions:

- The required programs have been keyed into your calculator.
- The stack is clear and you're starting with the default display format (FIX 4). Generally this does not affect the results of the example, except that your displays may not exactly match the ones in this book.
- The SIZE is set to 25 registers (the default). The number of storage registers needed (if any) are listed under "Remarks."

As you work the examples, also remember that lowercase letters are displayed as uppercase letters when they appear in menu labels.

If You Have a Printer. Many of the programs in this book will produce printed output if printing is enabled. Press ■[PRINT] [▲] PON to enable printing.

If you are not using a printer, be sure to disable printing (■[PRINT] [▲] POFF) to avoid losing results.

About Program Listings. It is assumed that you understand how to key programs into your calculator. If you're not sure, review part 2, "Programming," in the owner's manual.

If you print your programs, remember that the printer may print some characters differently than they are displayed. (For example, the → character is printed as ÷.) Also note that some printers cannot print the angle character (∡).

About the Subject Matter. Discussions on the various topics included are beyond the scope of this book. Refer to basic texts on the subjects of interest. Many references are available in university libraries and in technical and college bookstores. The examples in this book demonstrate approaches to solving problems, but they do not cover the many ways to approach general problems in mechanical engineering.

Our thanks to Dex Smith of TwentyEighth Street Publishing for developing this book.

1

Forces and Vectors

Vectors in the HP-42S

The HP-42S can represent a vector in two ways: as a complex number or a $1 \times n$ matrix. Of course, you can use matrices for all your work with vectors. However, when you are working with coplanar vectors, there are advantages to using complex numbers:

- You can see both parts (coordinates) of each complex number. (The contents of a matrix can only be viewed while editing the matrix.)
- You can enter and display complex numbers using rectangular or polar coordinates. (Vectors stored as matrices are always in rectangular coordinates regardless of the current coordinate mode.)

Using the "VEC" Program

The HP-42S provides most of the tools you will need for working with vectors. This program makes some of those tools easier to use by assigning them to the CUSTOM menu.

Remarks.

- The "VEC" program sets flag 27 to display the CUSTOM menu.
- Registers R_{00}, R_{01}, and R_{02} are used for temporary storage of the matrix elements. Therefore, the SIZE must be set to at least 3 storage registers (■[MODES] [▼] SIZE 3 [ENTER]).

Program Instructions.

1. Key the "VEC" program (listed on page 11) into your calculator.
2. Press [XEQ] VEC to execute the program.
3. Use the CUSTOM menu key assignments created by "VEC" for vector calculations.

   DOT calculates the dot product of the vectors in the X- and Y-registers.

   CROSS calculates the cross product of the vectors in the X- and Y-registers.

   UVEC calculates the unit vector of the vector in the X-register.

   ST→[ ] combines the coordinates in the X-, Y-, and Z-registers into a 1×3 matrix in the X-register.

   [ ]→ST returns the coordinates in a 1×3 matrix to the X-, Y-, and Z-registers.

   POL■ switches to Rectangular mode. (Press REC■ to return to Polar mode.)

Example. What is the unit vector of 10**i** −3**j** +17**k**?

[XEQ] VEC

```
x: 0.0000
DOT CROSS UVEC ST→[] []→ST POL■
```

To enter the *x*-, *y*-, and *z*-coordinates into the X-, Y-, and Z-registers (respectively), enter the *z*-coordinate first.

17 [ENTER] 3 [+/−] [ENTER] 10

```
x: 10_
DOT CROSS UVEC ST→[] []→ST POL■
```

Put the coordinates into vector (matrix) form.

ST→[ ]

```
x: [ 1x3 Matrix ]
DOT CROSS UVEC ST→[] []→ST POL■
```

Calculate the unit vector.

UVEC

```
x: [ 1x3 Matrix ]
DOT CROSS UVEC ST→[] []→ST POL■
```

Return the coordinates of the unit vector to the stack.

[ ]→ST

```
x: 0.5013
DOT CROSS UVEC ST→[] []→ST POL■
```

The *x*-coordinate of the unit vector is 0.5013 (to four decimal places).

[R↓]

```
x: -0.1504
DOT CROSS UVEC ST→[] []→ST POL■
```

The *y*-coordinate of the unit vector is −0.1504.

[R↓]

```
x: 0.8521
DOT CROSS UVEC ST→[] []→ST POL■
```

The *z*-coordinate of the unit vector is 0.8521.

"VEC" Program Listing.

Program: / **Comments:**

```
00 { 136-Byte Prgm }
01 LBL "VEC"
02 ASSIGN "DOT" TO 01
03 ASSIGN "CROSS" TO 02
04 ASSIGN "UVEC" TO 03
05 ASSIGN "ST→[]" TO 04
06 ASSIGN "[]→ST" TO 05
07 SF 27
08 LBL "REC■"
09 POLAR
10 ASSIGN "POL■" TO 06
11 RTN
12 LBL "POL■"
13 RECT
14 ASSIGN "REC■" TO 06
15 RTN
```

Makes CUSTOM menu key assignments.

```
16 LBL "ST→[]"
17 STO 00
18 R↓
19 STO 01
20 R↓
21 STO 02
22 R↓
```

Stores the X-, Y-, and Z-register values into registers 00-02.

```
23 3
24 ENTER
25 1
26 INDEX "REGS"
27 GETM
28 TRANS
29 RTN
```

Gets a 3 × 1 submatrix from the storage register matrix and transposes it to a 1 × 3 matrix.

```
30 LBL "[]→ST"
31 INDEX "REGS"
32 TRANS
33 PUTM
34 R↓
```

Transposes 1 × 3 matrix and puts it into the storage register matrix. Recalls the vector elements from registers 00-02.

```
35 RCL 02
36 RCL 01
37 RCL 00
38 END
```

Static Equilibrium at a Point

This program calculates the two reaction forces necessary to balance two-dimensional resultant force vectors. The direction of the reaction forces must be specified as an angle relative to the arbitrary axis.



$$R_1 \cos\theta_1 + R_2 \cos\theta_2 = \sum F \cos\phi$$

$$R_1 \sin\theta_1 + R_2 \sin\theta_2 = \sum F \sin\phi$$

Variables Used.

In Equations	Description	In Program
ΣF	Sum of forces.	ΣF
θ_1	Angle of first reaction force.	R1∡
θ_2	Angle of second reaction force.	R2∡

Remarks.

- Angles are entered and displayed using the current angular mode of the calculator.
- Since complex numbers are used for output, angles are normalized to produce positive magnitudes.
- A positive value of force (tension) points away from the origin; a negative value (compression) points toward the origin.
- The program produces an error if $R1∡ = R2∡$.
- This program sets Polar mode.
- Flags 21 and 55 are used to control printer output.

Program Instructions.

1. Key the "ΣF" program (listed on page 17) into your calculator.
2. Press [XEQ] ΣF to execute the program.
3. Repeat these steps for each force:
   a. Key in the magnitude of the force; press [ENTER].
   b. Key in the angle of the force (measured from the reference axis).
   c. Optional: press ■[COMPLEX]. (If you skip this step, the program automatically converts your inputs into a complex number.)
   d. Press ΣF+ to add the force. The current total force is displayed after each force is added.

If you enter a force incorrectly, you can delete it by adding an equal force in the opposite direction. If you notice the mistake immediately after entering it, press [+/-] ΣF+ .

4. Store the angles of the reaction forces:
   a. Key in $R1\measuredangle$ and then press R1∡ .
   b. Key in $R2\measuredangle$ and then press R2∡ .
5. Press R1R2 to calculate the reaction forces. (Press [◆] to clear the results from the display.)
6. To work another problem, press CLΣF to clear the accumulated forces and start over at step 3.

Example 1: Balancing a Single Vector. Find the reaction forces in the following diagram.



Set Degrees mode and run the "ΣF" program.

■[MODES] DEG
[XEQ] ΣF

```
ΣF=0.0000 ∡0.0000
ΣF+ CLΣF R1∡ R2∡      R1R2
```

Enter the force vector.

75 [ENTER] 135 ΣF+

```
ΣF=75.0000 ∡135.0000
ΣF+ CLΣF R1∡ R2∡      R1R2
```

Enter the angles of the reaction forces.

30 R1∡

```
R1∡=30.0000
ΣF+ CLΣF R1∡ R2∡      R1R2
```

270 R2∡

```
R2∡=270.0000
ΣF+ CLΣF R1∡ R2∡      R1R2
```

Now, calculate the reaction forces.

R1R2

```
R1=61.2372 ∡30.0000
R2=83.6516 ∡-90.0000
```

Example 2: Forces in a Bridge Truss. Find the reaction forces in structural members A and C. Assume pin connections at the joint.



Set Degrees mode and then start the program. (If you just worked the previous example, you do not have to restart the program. Press [←] to clear the message in the display and then press CLΣF to clear the forces accumulated above.)

■[MODES] DEG
[XEQ] ΣF

```
ΣF=0.0000 ∡0.0000
ΣF+ CLΣF R1∡ R2∡      R1R2
```

Accumulate the forces.

100 [ENTER] 45 ΣF+

```
ΣF=100.0000 ∡45.0000
ΣF+ CLΣF R1∡ R2∡      R1R2
```

120 [ENTER] 180 ΣF+

```
ΣF=86.1942 ∡124.8787
ΣF+ CLΣF R1∡ R2∡ R1R2
```

Enter the angles of the reaction forces.

0 R1∡

```
R1∡=0.0000
ΣF+ CLΣF R1∡ R2∡ R1R2
```

180 [ENTER] 45 [−] R2∡

```
R2∡=135.0000
ΣF+ CLΣF R1∡ R2∡ R1R2
```

Calculate the reaction forces.

R1R2

```
R1=21.4214 ∡180.0000
R2=100.0000 ∡-45.0000
```

Note that for this problem, the reaction forces are displayed in the opposite direction of the angles entered for R_1 and R_2. This indicates that the forces are *compressive*.

"ΣF" Program Listing.

Program:	**Comments:**
00 { 245-Byte Prgm }	
01 LBL "ΣF" 02 CF 21 03 FS? 55 04 SF 21 05 CLMENU 06 POLAR	Sets or clears flag 21 to match flag 55. Clears the programmable menu and sets Polar mode.
07 XEQ 02	Clears the stack and variables.
08 "ΣF+" 09 KEY 1 XEQ 01 10 "CLΣF" 11 KEY 2 XEQ 02 12 "R1∡" 13 KEY 3 XEQ 03	Defines the menu keys.

14 "R2∡" 15 KEY 4 XEQ 04 16 "R1R2" 17 KEY 6 XEQ 06 18 KEY 9 GTO 09	
19 LBL 00 20 MENU 21 STOP 22 GTO 00	Displays the menu and stops. Pressing [R/S] redisplays the menu.
23 LBL 01 24 REAL? 25 COMPLEX 26 STO+ "ΣF" 27 VIEW "ΣF" 28 RTN	Adds the input to Σ*F*. If the input is not already complex, it's converted automatically.
29 LBL 02 30 CLST 31 STO "R1∡" 32 STO "R2∡" 33 COMPLEX 34 STO "ΣF" 35 VIEW "ΣF" 36 RTN	Clears the stack and initializes the variables.
37 LBL 03 38 CPX? 39 COMPLEX 40 STO "R1∡" 41 VIEW "R1∡" 42 RTN	Stores and displays *R1∡*. (*R1∡* cannot be complex.)
43 LBL 04 44 CPX? 45 COMPLEX 46 STO "R2∡" 47 VIEW "R2∡" 48 RTN	Stores and displays *R2∡*. (*R2∡* cannot be complex.)

Calculates R_1.

```
49 LBL 06
50 RCL "ΣF"
51 RECT
52 COMPLEX
53 POLAR
54 RCL "R2∡"
55 SIN
56 STO× ST Z
57 LASTX
58 COS
59 STO× ST Z
60 R↓
61 R↓
62 -
63 RCL "R1∡"
64 COS
65 STO× ST T
66 R↓
67 LASTX
68 SIN
69 RCL× ST Z
70 R↑
71 -
72 STO ST Z
73 ÷
74 RCL "R1∡"
75 COMPLEX
76 "R1="
77 ARCL ST X
```

Calculates R_2.

```
78 X<>Y
79 RCL "ΣF"
80 RECT
81 COMPLEX
82 POLAR
83 RCL "R1∡"
84 SIN
85 STO× ST Z
86 R↓
```

87 LASTX	
88 COS	
89 ×	
90 X<>Y	
91 -	
92 X<>Y	
93 ÷	
94 RCL "R2∡"	
95 COMPLEX	
96 ⊢"␊R2="	
97 ARCL ST X	
98 AVIEW	Displays the results.
99 RTN	
100 LBL 09	Exits all menus and ends.
101 EXITALL	
102 END	

2

Equations of Motion

Equations of Motion

This chapter contains three Solver programs for the following equations of motion for constant acceleration.

$$s = v_o t + 1/2at^2$$

$$v = v_o + at$$

$$v^2 = v_o^2 + 2as$$

Variables Used.

In Equations	Description	In Programs
t	Time.	t
a	Acceleration.	a
v_o	Initial velocity.	v0
v	Final velocity.	v
s	Displacement.	s

Remarks. These Solver programs can also be used for angular motion. Use a, v_o, v, and s to represent α, ω_o, ω, and θ, respectively.

Program Instructions.

1. Key the appropriate program ("MO1", "MO2", or "MO3") into your calculator. (The programs are listed on pages 24 and 25.)
2. Press ■[SOLVER].
3. Select "MO1", "MO2", or "MO3" by pressing the corresponding menu key.
4. Use the variable menu to store the known values. Key in each value and then press the corresponding menu key.

5. Optional: Store guesses into the unknown variable to direct the Solver to a particular solution (there might be more than one).
6. Press the menu key for the unknown value. The Solver immediately begins to search for a solution.

Example. If a rock dropped off a bridge hits the water 1.6 seconds later, how high is the bridge? What is the velocity of the rock when it strikes the water?

Since each of the equations only uses four of the five variables, many problems with two unknowns can be solved by carefully choosing the programs you use. In this case, *s* and *v* are unknown. First solve for *s* using the equation that does not require *v* ("MO1"). Then use one of the other two equations to find *v*. (Key "MO1" and "MO2" into your calculator before you begin this example.)

■ [SOLVER] MO1	x: 0.0000 T A VO S

Store the known values: t = 1.6 seconds, a = -9.8 m/s^2 (for free falling objects), and v_0 = 0.

1.6 T	t=1.6000 T A VO S
9.8 [+/-] A	a=-9.8000 T A VO S
0 V0	v0=0.0000 T A VO S
S	s=-12.5440 T A VO S

The bridge is about 12.5 meters above the water. Now find *v*. Since *s* is now known, you can use "MO2" or "MO3".

[EXIT] MO2

```
x: -12.5440
 T  A  V0  V
```

When you first select a Solver program (as you just have), the first menu key you press executes a *store* to that variable, even if you don't key in a number first. Therefore, you'll have to press V twice to start the Solver.

V

```
v=-12.5440
 T  A  V0  V
```

V

```
v=-15.6800
 T  A  V0  V
```

The rock is traveling at almost 15.7 meters per second when it hits the water.

"MO1" Program Listing.

```
00 { 40-Byte Prgm }
01 LBL "MO1"
02 MVAR "t"
03 MVAR "a"
04 MVAR "v0"
05 MVAR "s"
06 RCL "t"
07 ENTER
08 X↑2
09 RCLx "a"
10 2
11 ÷
12 X<>Y
13 RCLx "v0"
14 +
15 RCL- "s"
16 END
```

“MO2” Program Listing.

```
00 { 33-Byte Prgm }
01 LBL "MO2"
02 MVAR "t"
03 MVAR "a"
04 MVAR "v0"
05 MVAR "v"
06 RCL "t"
07 RCL× "a"
08 RCL+ "v0"
09 RCL- "v"
10 END
```

“MO3” Program Listing.

```
00 { 39-Byte Prgm }
01 LBL "MO3"
02 MVAR "a"
03 MVAR "v0"
04 MVAR "v"
05 MVAR "s"
06 RCL "v"
07 X↑2
08 RCL "v0"
09 X↑2
10 -
11 2
12 RCL× "a"
13 RCL× "s"
14 -
15 END
```

3

Analysis Programs

Mohr Circle Analysis

This program calculates the 2-D Mohr's circle for stress using strain gage data (equiangular or rectangular) or using known stresses.

Configuration Code	**1**	**2**
Type of Rosette	Rectangular	Delta (Equiangular)
		
Principal Strains ϵ_1, ϵ_2	$\frac{1}{2}\left[\epsilon_a + \epsilon_c \pm \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2}\right]$	$\frac{1}{3}\left[\epsilon_a + \epsilon_b + \epsilon_c \pm \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2 + 2(\epsilon_c - \epsilon_a)^2}\right]$
Center of Mohr Circle $\frac{s_1 + s_2}{2}$	$\frac{E(\epsilon_a + \epsilon_c)}{2(1 - v)}$	$\frac{E(\epsilon_a + \epsilon_b + \epsilon_c)}{3(1 - v)}$
Maximum Shear Stress τ_{max}	$\frac{E}{2(1 + v)} \times \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2}$	$\frac{E}{3(1 + v)} \times \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2 + 2(\epsilon_c - \epsilon_a)^2}$
Orientation of Principal Stresses ϕ	$\frac{1}{2}\tan^{-1}\left[\frac{2\epsilon_b - \epsilon_a - \epsilon_c}{\epsilon_a - \epsilon_c}\right]$	$\frac{1}{2}\tan^{-1}\left[\frac{\sqrt{3}\,(\epsilon_c - \epsilon_b)}{(2\epsilon_a - \epsilon_b - \epsilon_c)}\right]$

Variables Used.

In Equations	Description	In Program
E	Young's modulus.	E
ν	Poisson's ratio.	V
ε_a	Strain ε_0.	e1
ε_b	Strain ε_{45} or ε_{60}.	e2
ε_c	Strain ε_{90} or ε_{120}.	e3
σ_x	Normal stress on the *x*-face.	X
σ_y	Normal stress on the *y*-face.	Y
σ_1	Minimum principal stress.	Smin
σ_2	Maximum principal stress.	Smax
τ_{xy}	Shear stress.	TAUxy
τ_{max}	Maximum shear stress.	TAUmax
ϕ	Counterclockwise angle from ε_a to the maximum principal stress.	∡
θ	Counterclockwise angle from the *x*-axis to the maximum principal axis.*	∡
θ'	Arbitrary angle counter- clockwise from the *x*-axis.*	ARB∡
S	Normal stress.*	
τ	Shear stress.*	

*Refer to the equations on page 32.

Remarks.

- Shearing stress on any element is positive if it tends to rotate the element clockwise. Tensile forces are considered positive; compressive forces are considered negative.
- Angles are entered and displayed using the current angular mode of the calculator.
- The program calculates the principal stresses for a two-dimensional stress state only. However, a knowledge of the stresses in the *z*-direction is necessary to determine the overall maximum and minimum stresses.
- This program uses storage register R_{00} to store intermediate results. Before running this program be sure the SIZE is set to at least 1 (■[MODES] [▼] SIZE 1 [ENTER]).
- The program sets flag 21, which causes the program to stop each time a result is displayed if printing is disabled.

Program Instructions.

1. Key the "MOHR" program (listed on page 34) into your calculator.
2. Press [XEQ] MOHR to execute the program.
3. When the program displays Type of Input?, press *one* of the following:

   EQ∠ if equiangular strain gage readings are known.

   REC if rectangular strain gage readings are known.

   STRES if stresses are known directly.
4. The program then displays Store Values; [R/S] and a variable menu containing the variables to be entered. Store each variable by keying in the value and then pressing the corresponding menu key. After all of the values have been stored, press [R/S]. The program then calculates and displays the results. If you are not using a printer, press [R/S] after each result.

5. Optional: To calculate the normal stress and the shear stress, do the following:
   a. When you see ARB∡?, key in the arbitrary angle of rotation.
   b. Press [R/S]. If you are not using a printer, press [R/S] after each result.

Example 1: Equiangular Strain Gage. An equiangular rosette strain gage measures the following strains:

$\varepsilon_0 = 180\ \mu$

$\varepsilon_{60} = 200\ \mu$

$\varepsilon_{120} = -290\ \mu$

Find the principal stresses and their orientation. The material properties are $E = 30 \times 10^6$ psi and $\nu = 0.3$.

■[MODES] DEG [XEQ] MOHR	Type of Input? EQ∡ | REC | STRES
EQ∡	Store Values; [R/S] E | V | E1 | E2 | E3
30 [E] 6 E	E=30,000,000.0000 E | V | E1 | E2 | E3
.3 V	V=0.3000 E | V | E1 | E2 | E3
180 [E] 6 [+/-] E1	e1=0.0002 E | V | E1 | E2 | E3

200 [E] 6 [+/-] E2 | e2=0.0002

290 [+/-] [E] 6 [+/-] E3 | e3=-0.0003

[R/S] | Smax=8,675.1358 x: 8,675.1358

[R/S] | Smin=-6,103.7072 x: -6,103.7072

[R/S] | TAUmax=7,389.4215 x: -6,103.7072

[R/S] | ∡=31.0333 x: -6,103.7072

Example 2: Known Stresses. The stresses acting on an element are shown below (all stresses are in MPa).



$$\tau_{max} = \left[\left(\frac{S_x - S_y}{2}\right)^2 + \tau_{xy}^2\right]^{\frac{1}{2}}$$

$$S_1 = \frac{S_x + S_y}{2} + \tau_{max}$$

$$S_2 = \frac{S_x + S_y}{2} - \tau_{max}$$

$$\theta = \frac{1}{2}\tan^{-1}\left(\frac{-2\tau_{xy}}{S_x - S_y}\right)$$

$$S = \frac{S_1 + S_2}{2} + \tau_{max}\cos 2\left(\theta' - \theta\right)$$

$$\tau = \tau_{max}\sin 2\left(\theta' - \theta\right)$$

Find the principal stresses and their orientation, and the stresses on the face of the element oriented 45° counterclockwise from the x-axis.

■ [MODES] DEG [XEQ] MOHR	Type of Input? EQ∡ | REC | STRES
STRES	Store Values; [R/S] TAUXY | X | Y
30 TAUXY	TAUxy=30.0000 TAUXY | X | Y
75 X	X=75.0000 TAUXY | X | Y

20 [+/-] Y	Y=-20.0000 TAUXY X Y
[R/S]	Smax=83.6805 x: 83.6805
[R/S]	Smin=-28.6805 x: -28.6805
[R/S]	TAUmax=56.1805 x: -28.6805
[R/S]	∡=-16.1378 x: -28.6805
[R/S]	Y: -28.6805 ARB∡?0.0000
45 [R/S]	Norm=-2.5000 x: -2.5000
[R/S]	Shear=47.5000 x: 47.5000

“MOHR” Program Listing.

Program:	Comments:
`00 { 412-Byte Prgm }`	
`01 LBL "MOHR"` `02 SF 21` `03 CLMENU` `04 "EQ∡"` `05 KEY 1 XEQ 01` `06 "REC"` `07 KEY 2 XEQ 02` `08 "STRES"` `09 KEY 3 XEQ 03` `10 KEY 9 GTO 09`	Sets flag 21 to control output and defines the menu for selecting the type of analysis.
`11 LBL A` `12 "Type of Input?"` `13 MENU` `14 PROMPT` `15 GTO A`	Displays the menu and prompts for the type of input. Pressing [R/S] redisplays the menu.
`16 LBL "M1"` `17 MVAR "E"` `18 MVAR "V"` `19 MVAR "e1"` `20 MVAR "e2"` `21 MVAR "e3"`	Declares the menu variables for the first variable menu.
`22 LBL "M2"` `23 MVAR "TAUxy"` `24 MVAR "X"` `25 MVAR "Y"`	Declares the menu variables for the second variable menu.
`26 LBL 00` `27 VARMENU "M1"` `28 XEQ 08`	Selects the first variable menu for input.
`29 RCL "E"` `30 1` `31 RCL- "V"`	Calculates τ_{max} for strain gage data.

```
32 ÷
33 STO 00
34 RCL "E"
35 1
36 RCL+ "V"
37 ÷
38 STO "TAUmax"
39 RCL "e1"
40 RCL "e2"
41 RCL "e3"
42 RTN
```

```
43 LBL 01
44 XEQ 00
45 -
46 3
47 SQRT
48 ÷
49 RCL "e2"
50 RCL+ "e3"
51 2
52 ×
53 RCL- ST Z
54 3
55 ÷
56 RCL ST Z
57 GTO 04
```

Calculates results using equiangular strain gage readings.

```
58 LBL 02
59 XEQ 00
60 RCL+ ST Z
61 2
62 ÷
63 RCL- ST Y
64 RCL "e3"
65 R↑
66 GTO 04
```

Calculates results using rectangular strain gage readings.

```
67 LBL 03
68 VARMENU "M2"
69 XEQ 08
```

Selects the second variable menu for input.

```
70 1
71 STO 00
72 STO "TAUmax"
73 RCL "TAUxy"
74 +/-
75 RCL "Y"
76 RCL "X"
```

Calculates results using known stresses.

```
77 LBL 04
78 STO "Smin"
79 +
80 2
81 ÷
82 STO× 00
83 RCL- "Smin"
84 ABS
85 →POL
86 STO× "TAUmax"
87 X<>Y
88 2
89 ÷
90 STO "∡"
```

Calculates τ_{max} and θ.

```
91 RCL "TAUmax"
92 RCL+ 00
93 STO "Smax"
94 VIEW "Smax"
95 RCL 00
96 RCL- "TAUmax"
97 STO "Smin"
98 VIEW "Smin"
```

Calculates S_{max} and S_{min}.

```
99 VIEW "TAUmax"
100 VIEW "∡"
```

Displays τ_{max} and θ.

```
101 INPUT "ARB∡"
102 RCL- "∡"
103 2
104 ×
105 SIN
```

Optional routine calculates normal and shear stress at an arbitrary angle.

```
106 LASTX
107 COS
108 RCL× "TAUmax"
109 RCL+ 00
110 "Norm="
111 ARCL ST X
112 AVIEW
113 R↓
114 RCL× "TAUmax"
115 "Shear="
116 ARCL ST X
117 AVIEW
118 RTN
```

```
119 LBL 08
120 "Store Values; "
121 ├"[R/S]"
122 PROMPT
123 LBL 09
124 EXITALL
125 END
```

Prompts for input using the selected variable menu. Pressing [R/S] clears the menu and continues.

Soderberg's Equation for Fatigue

This program is used with the Solver to calculate any one of the six variables in Soderberg's equation for fatigue when the other five are known. Soderberg's equation is shown graphically in the following figure.



$$\frac{S_{yp}}{FS} = \frac{1}{2}\left[s_{max} + s_{min} + K\left(\frac{S_{yp}}{S_e}\right)\left(s_{max} - s_{min}\right)\right]$$

Variables Used.

In Equation	Description	In Program
s_{yp}	Yield point stress.	Syp
s_e	Endurance stress.	Se
s_{max}	Maximum applied stress.	Smax
s_{min}	Minimum applied stress.	Smin
K	Stress concentration factor.	K
FS	Factor of safety.	FS

Remarks.

- Soderberg's equation is valid for ductile materials only.
- Fatigue effects are magnified in corrosive environments.

Program Instructions.

1. Key the "SODER" program (listed on page 41) into your calculator.
2. Press ■[SOLVER] SODER to select the Solver equation.
3. Store each of the known variables by keying in the value and then pressing the corresponding menu key.
4. Press the menu key for the unknown variable.

Example. Given the following values, what is the maximum allowable applied stress if the minimum applied stress is 15,000 psi?

s_{yp} = 80,000 psi

s_e = 30,000 psi

K = 1.5

FS = 2.0

■ [SOLVER] SODER	x: 0.0000 SYP SE SMAX SMIN K FS
80000 SYP	Syp=80,000.0000 SYP SE SMAX SMIN K FS
30000 SE	Se=30,000.0000 SYP SE SMAX SMIN K FS
15000 SMIN	Smin=15,000.0000 SYP SE SMAX SMIN K FS
1.5 K	K=1.5000 SYP SE SMAX SMIN K FS
2 FS	FS=2.0000 SYP SE SMAX SMIN K FS
SMAX	Smax=25,000.0000 SYP SE SMAX SMIN K FS

"SODER" Program Listing.

```
00 { 82-Byte Prgm }
01 LBL "SODER"
02 MVAR "Syp"
03 MVAR "Se"
04 MVAR "Smax"
05 MVAR "Smin"
06 MVAR "K"
07 MVAR "FS"
08 RCL "Smax"
09 RCL+ "Smin"
10 LASTX
11 RCL- "Smin"
12 RCL× "K"
13 RCL× "Syp"
14 RCL÷ "Se"
15 +
16 2
17 ÷
18 RCL "Syp"
19 RCL÷ "FS"
20 -
21 END
```

Composite Section Properties

The mathematical properties of a cross section composed of a finite number of rectangular elements can be computed by adding the contribution of each rectangular region individually. This program uses this principle to calculate the area of a section, the moments of inertia about the specified set of axes, the moments of inertia about an axis translated to the centroid, the moments of inertia of the principal axes, and the angle of rotation between the translated axes and the principal axis.

$$A_{si} = \Delta x_i \Delta y_i$$

$$A = A_{s1} + A_{s2} + A_{s3} + \cdots + A_{sn}$$

$$\overline{x} = \frac{\sum_{i=1}^{n} x_{0i} A_{si}}{A}$$

$$\overline{y} = \frac{\sum_{i=1}^{n} y_{0i} A_{si}}{A}$$

$$I_{xy} = \sum_{i=1}^{n} x_{0i} y_{0i} A_{si} \qquad I_{\overline{x}\,\overline{y}} = I_{xy} - A\,\overline{x}\,\overline{y}$$

$$I_x = \sum_{i=1}^{n} \left(y_{0i}^2 + \frac{\Delta y_i^2}{12} \right) A_{si} \qquad I_{\overline{x}} = I_x - A\,\overline{y}^2$$

$$I_y = \sum_{i=1}^{n} \left(x_{0i}^2 + \frac{\Delta x_i^2}{12} \right) A_{si} \qquad I_{\overline{y}} = I_y - A\,\overline{x}^2$$

$$J = I_x + I_y$$

$$\phi = \frac{1}{2}\tan^{-1}\left(\frac{2I_{\overline{x}\,\overline{y}}}{I_{\overline{x}} - I_{\overline{y}}}\right)$$

$$I_{\overline{x}\phi} = I_{\overline{x}}\cos^2\phi + I_{\overline{y}}\sin^2\phi + I_{\overline{x}\,\overline{y}}\sin 2\phi$$

$$I_{\overline{y}\phi} = I_{\overline{y}}\cos^2\phi + I_{\overline{x}}\sin^2\phi + I_{\overline{x}\,\overline{y}}\sin 2\phi$$

$$J_{\phi} = I_{\overline{x}\phi} + I_{\overline{y}\phi}$$

Variables and Storage Registers Used.

In Equations	Description	In Program
Δx_i	Width of a rectangular element.	W
Δy_i	Height of a rectangular element.	H
A_{si}	Area of an element.	R_{02}
A	Total area of the section.	R_{01}
$\overline{x}$	*x*-coordinate of the centroid (total).	$R_{03} \div R_{01}$
$\overline{y}$	*y*-coordinate of the centroid (total).	$R_{04} \div R_{01}$
x_{0i}	*x*-coordinate of the centroid of an element.	Xc
y_{0i}	*y*-coordinate of the centroid of an element.	Yc
I_x	Moment of inertia about the x-axis.	R_{06}
I_y	Moment of inertia about the y-axis.	R_{07}
J	Polar moment of inertia about the origin.	R_{08}
I_{xy}	Product of inertia about the origin.	R_{09}

In Equations	Description	In Program
$I_{\bar{x}}$	Moment of inertia about the x-axis translated to the centroid.	R_{06}
$I_{\bar{y}}$	Moment of inertia about the y-axis translated to the centroid.	R_{07}
$I_{\overline{xy}}$	Product of inertia about the translated axis.	R_{09}
ϕ	Angle between the translated axis and the principal axis.	R_{05}
$I_{\bar{x}\phi}$	Moment of inertia about the principal x-axis.	
$I_{\bar{y}\phi}$	Moment of inertia about the principal y-axis.	
J_{ϕ}	Polar moment of inertia about the principal axis.	R_{08}

Remarks.

- Angles are entered and displayed using the current angular mode of the calculator.
- For a given origin, the polar moment of inertia is constant regardless of the angular rotation. Therefore, $I_{\overline{xy}}$ is equal to J_{ϕ}
- It is possible to obtain a negative value for the product of inertia.
- Before using "SECTION" be sure to set the SIZE to at least 11 registers (■[MODES] [▼] SIZE 11 [ENTER]). The table above shows that some of the registers are used more than once. Storage registers not listed in the table are used for other intermediate results.
- This program clears *all* of the storage registers.
- Results are labeled and displayed in the same order as in the table above. The program uses variables only for the four section inputs.
- This program uses flags 21 and 55 to control printer output.

Program Instructions.

1. Key the "SECTION" program (listed on page 53) into your calculator.
2. Press [XEQ] SECTI to execute the program.
3. When you see Number of Sections?, key in the number of composite sections and then press [R/S].
4. Enter the data for each section:
   a. Key in the *x*-coordinate of the centroid and then press XC .
   b. Key in the *y*-coordinate of the centroid and then press YC .
   c. Key in the width of the section and then press W .
   d. Key in the height of the section and then press H .
   e. Press [R/S] to continue.

   If you store an incorrect value, simply repeat the step to store the correct value. If you notice that you've made a mistake after pressing [R/S], start over at step 2.
5. After the data for each section has been entered, the program calculates the results. If you are not using a printer, press [R/S] after each result is displayed.

Example 1: Rectangular Section. Calculate the section properties of the following cross section.



Table of Inputs

x_c	y_c	Δx	Δy
2	1.5	4	3

Keys	Display
■ [MODES] DEG [XEQ] SECTI	Number of Sections? x: 0.0000
1 [R/S]	Input Section 1;[R/S] XC YC W H
2 XC	Xc=2.0000 XC YC W H
1.5 YC	Yc=1.5000 XC YC W H

Keys	Display
4 W	W=4.0000 XC YC W H
3 H	H=3.0000 XC YC W H
[R/S]	Area= x: 12.0000
[R/S]	Xc= x: 2.0000
[R/S]	Yc= x: 1.5000
[R/S]	Ix= x: 36.0000
[R/S]	Iy= x: 64.0000
[R/S]	J= x: 100.0000
[R/S]	Ixy= x: 36.0000

[R/S]	Ixc= x: 9.0000
[R/S]	Iyc= x: 16.0000
[R/S]	Ixyc= x: 0.0000
[R/S]	∡phi= x: 0.0000
[R/S]	Ixc phi= x: 9.0000
[R/S]	Iyc phi= x: 16.0000
[R/S]	Jphi= x: 25.0000

Example 2: Composite Section. Calculate the section properties of the following section.



Table of Inputs

	x_c	y_c	Δx	Δy
1	5	11.5	6	1
2	1	6	2	12
3	7	1	10	2

■[MODES] DEG
[XEQ] SECTI

```
Number of Sections?
x: 25.0000
```

3 [R/S]

```
Input Section 1;[R/S]
XC | YC | W | H | |
```

5 XC

```
Xc=5.0000
XC | YC | W | H | |
```

11.5 YC

```
Yc=11.5000
XC | YC | W | H | |
```

6 W	W=6.0000
1 H	H=1.0000
[R/S]	Input Section 2;[R/S]
1 XC	Xc=1.0000
6 YC	Yc=6.0000
2 W	W=2.0000
12 H	H=12.0000
[R/S]	Input Section 3;[R/S]
7 XC	Xc=7.0000

1 YC	Yc=1.0000 XC YC W H
10 W	W=10.0000 XC YC W H
2 H	H=2.0000 XC YC W H
[R/S]	Area= x: 50.0000
[R/S]	Xc= x: 3.8800
[R/S]	Yc= x: 4.6600
[R/S]	Ix= x: 1,972.6667
[R/S]	Iy= x: 1,346.6667
[R/S]	J= x: 3,319.3333

[R/S]	Ixy= x: 629.0000
[R/S]	Ixc= x: 886.8867
[R/S]	Iyc= x: 593.9467
[R/S]	Ixyc= x: -275.0400
[R/S]	∡phi= x: 30.9814
[R/S]	Ixc phi= x: 1,052.0261
[R/S]	Iyc phi= x: 428.8072
[R/S]	Jphi= x: 1,480.8333

"SECTION" Program Listing.

Program:	Comments:
00 { 395-Byte Prgm }	
01 LBL "SECTION" 02 MVAR "Xc" 03 MVAR "Yc" 04 MVAR "W" 05 MVAR "H"	Defines the menu variables.
06 LBL 00 07 "Number of " 08 ⊢"Sections?" 09 PROMPT 10 ABS 11 IP 12 X=0? 13 GTO 00	Prompts for the number of sections; rejects a zero input.
14 CLRG 15 1E3 16 ÷ 17 1 18 + 19 STO 00	Clears the storage registers and stores a loop counter in R_{00}.
20 LBL 01 21 VARMENU "SECTION"	Selects the variable menu.
22 "Input Section " 23 RCL 00 24 AIP 25 ⊢";[R/S]"	Builds the prompt string using the loop counter.
26 CF 21 27 AVIEW 28 CLA 29 SF 21 30 STOP 31 ALENG	Displays the prompt and stops. Pressing [R/S] continues the program.

```
32 X≠0?
33 GTO 01
```

```
34 EXITALL
35 RCL "H"
36 RCL× "W"
37 STO 02
38 STO+ 01
39 RCL× "Yc"
40 STO+ 04
41 RCL "Xc"
42 RCL× 02
43 STO+ 03
44 RCL "Yc"
45 X↑2
46 RCL "H"
47 X↑2
48 12
49 ÷
50 +
51 RCL× 02
52 STO+ 06
53 RCL "Xc"
54 X↑2
55 RCL "W"
56 X↑2
57 12
58 ÷
59 +
60 RCL× 02
61 STO+ 07
62 +
63 STO+ 08
64 RCL "Xc"
65 RCL× "Yc"
66 RCL× 02
67 STO+ 09
```

Accumulates the inputs into the appropriate storage registers. (Refer to the table starting on page 43.)

```
68 ISG 00
69 GTO 01
```

Repeats loop for each section.

70 RCL 09 71 RCL 03 72 RCL× 04 73 RCL÷ 01 74 - 75 STO 10 76 RCL 03	Calculates $I_{\overline{xy}}$.
77 RCL 01 78 "Area" 79 XEQ 03	Displays the area.
80 ÷ 81 "Xc" 82 XEQ 03	Calculates and displays $\overline{x}$.
83 X↑2 84 RCL 04 85 RCL÷ 01 86 "Yc" 87 XEQ 03	Calculates and displays $\overline{x}$.
88 X↑2 89 RCL× 01 90 +/- 91 RCL 06 92 "Ix" 93 XEQ 03 94 + 95 STO 06	Displays I_x and calculates $I_{\overline{x}}$.
96 R↓ 97 RCL× 01 98 +/- 99 RCL 07 100 "Iy" 101 XEQ 03	Displays I_y.
102 RCL 08 103 "J" 104 XEQ 03	Displays J.

105 R↓ 106 RCL 09 107 "Ixy" 108 XEQ 03	Displays I_{xy}.
109 R↓ 110 + 111 STO 07	Calculates $I_{\bar{y}}$.
112 RCL 06 113 "Ixc" 114 XEQ 03	Displays $I_{\bar{x}}$.
115 STO 08 116 X<>Y 117 "Iyc" 118 XEQ 03	Displays $I_{\bar{y}}$.
119 STO+ 08 120 - 121 STO 04 122 RCL 10 123 "Ixyc" 124 XEQ 03	Displays $I_{\overline{xy}}$.
125 X=0? 126 GTO 02 127 ÷ 128 1/X 129 2 130 × 131 ATAN 132 2 133 ÷ 134 +/- 135 LBL 02 136 STO 05 137 "∡phi" 138 XEQ 03	Calculates ϕ.

```
139 2
140 ×
141 SIN
142 RCL× 10
143 +/-
144 RCL 05
145 SIN
146 X↑2
147 RCL× 04
148 -
149 RCL+ 06
150 "Ixc phi"
151 XEQ 03
```

Calculates and displays $I_{\bar{x}\phi}$.

```
152 RCL- 08
153 +/-
154 "Iyc phi"
155 XEQ 03
```

Calculates and displays $I_{\bar{y}\phi}$.

```
156 RCL 08
157 "Jphi"
158 XEQ 03
```

Displays J_{ϕ}.

```
159 GTO 00
```

Returns to the initial prompt.

```
160 LBL 03
161 ⊢"="
162 FS? 55
163 ARCL ST X
164 AVIEW
165 END
```

Displays a result. If results are being printed (flag 55 set), the label and value are printed on a single line.

Spring Constant

This Solver program calculates the value of any of the five variables in the following spring equation, given the other four.

$$k = \frac{F_1 - F_2}{X_2 - X_1}$$

Variables Used.

In Equations	Description	In Program
X_1	First spring length.	X1
F_1	Force required to retain spring at length X_1.	F1
X_2	Second spring length.	X2
F_2	Force required to retain spring at length X_2.	F2
k	Spring constant.	k

Remarks. The "SPRING" Solver program can also be used to solve any general linear equation of the form $y - y_0 = m(x - x_0)$.

Program Instructions.

1. Key the "SPRING" program (listed on page 60) into your calculator.
2. Press ■[SOLVER] SPRIN.
3. Use the variable menu to store each of the known values by keying in the value and then pressing the corresponding menu key.
4. Press the menu key for the unknown.

Example. A compression spring is 4.0 cm long under no compressive forces. A force of 27 Newtons compresses the spring to a length of 2.8 cm. Find the spring constant, *k*, if the solid (fully compressed) height of the spring is 2.5 cm. Also find the force required to fully compress the spring.

■[SOLVER] SPRIN	x: 0.0000 X1 F1 X2 F2 K
4 X1	X1=4.0000 X1 F1 X2 F2 K
0 F1	F1=0.0000 X1 F1 X2 F2 K
2.8 X2	X2=2.8000 X1 F1 X2 F2 K
27 F2	F2=27.0000 X1 F1 X2 F2 K
K	k=22.5000 X1 F1 X2 F2 K

The spring constant is 22.5.

2.5 X2	X2=2.5000 X1 F1 X2 F2 K
F2	F2=33.7500 X1 F1 X2 F2 K

A force of 33.75 Newtons would completely compress the spring (to 2.5 cm).

"SPRING" Program Listing.

```
00 { 49-Byte Prgm }          07 RCL "F1"
01 LBL "SPRING"              08 RCL- "F2"
02 MVAR "X1"                 09 RCL "X2"
03 MVAR "F1"                 10 RCL- "X1"
04 MVAR "X2"                 11 ÷
05 MVAR "F2"                 12 RCL- "k"
06 MVAR "k"                  13 END
```

Linear or Angular Deformation

This program solves for linear deflection under tensile load, or the analogous angular deflection under torque. Given four of the five variables, the unknown is calculated.



Variables Used. Since the form of the two equations is identical, either type of deformation problem can be solved using this program. Use the following table to relate the variables used in the equations to the variables used in the program.

In Equations	Description	In Program
Δl	Change in length.	Delta
P	Applied load.	P
l	Length.	L
A	Cross-sectional area.	A
E	Modulus of elasticity.	E
θ	Deflection angle (in radians).	Delta
T	Applied torque.	P
J	Polar moment of the section.	A
G	Modulus of elasticity in shear.	E

Remarks.

- This program is not applicable for non-elastic media or elastic media where stress exceeds the elastic limit. Materials must be isotropic. The equation for angular deflection is not valid in the neighborhood of the applied torque.
- Even though the polar moment must be expressed in radians, it is not necessary to set Radians mode because no trigonometric functions are used in the program.

Program Instruction.

1. Key the "DEFORM" program (listed on page 64) into your calculator.
2. Press ■[SOLVER] DEFOR to select the program.
3. Store each of the known values by keying in the value and then pressing the corresponding menu key.

4. Solve for the unknown value by pressing the corresponding menu key.

Example. Steel bars (each 10 meters long) affixed to a roof are to be used to support the end of a cantilever balcony. The load on each bar will be 50,000 N. If the maximum allowable deflection is 0.001 meters, what should the area of the bars be? The modulus of elasticity is given as 2.068×10^{11} N/m^2.

■ [SOLVER] DEFOR	x: 0.0000 DELTA P L A E
.001 DELTA	Delta=0.0010 DELTA P L A E
50000 P	P=50,000.0000 DELTA P L A E
10 L	L=10.0000 DELTA P L A E
2.068 [E] 11 E	E=206,800,000,000. DELTA P L A E
A	A=0.0024 DELTA P L A E

“DEFORM” Program Listing.

```
{ 48-Byte Prgm }
LBL "DEFORM"
MVAR "Delta"
MVAR "P"
MVAR "L"
MVAR "A"
MVAR "E"
RCL "P"
RCL× "L"
RCL÷ "A"
RCL÷ "E"
RCL- "Delta"
END
```

4

Thermodynamics

Equations of State

This section contains programs for both ideal gas and Redlich-Kwong equations of state.

The ideal gas equation is

$$PV = nRT$$

The Redlich-Kwong equation is

$$P = \frac{nRT}{(V - b)} - \frac{a}{T^{1/2} V (V + b)}$$

where:

$$a = 4.934\, b\; nRT_c^{1.5}$$

$$b = 0.0867 \frac{nRT_c}{P_c}$$

Variables Used.

In Equations	Description	In Program
P	Absolute pressure.	P
V	Volume.	V
n	Number of moles present.	n
R	Universal gas constant.	R
T	Absolute temperature.	T
T_c	Critical temperature.	Tc
P_c	Critical pressure.	Pc

Values of the Universal Gas Constant

Value of R	Units of R	Units of P	Units of V	Units of T
8.314	N-m/g mole-K	N/m^2	m^3/g mole	K
83.14	cm^3-bar/g mole-K	bar	cm^3/g mole	K
82.05	cm^3-atm/g mole-K	atm	cm^3/g mole	K
0.7302	atm-ft^3/lb mole-°R	atm	ft^3/lb mole	°R
10.73	psi-ft^3/lb mole-°R	psi	ft^3/lb mole	°R
1545	psf-ft^3/lb mole-°R	psf	ft^3/lb mole	°R

Critical Temperatures and Pressures

Substance	T_c, K	T_c, °R	P_c, atm
Ammonia	405.6	730.1	112.5
Argon	151	272	48.0
Carbon dioxide	304.2	547.6	72.9
Carbon monoxide	133	239	34.5
Chlorine	417	751	76.1
Helium	5.3	9.5	2.26
Hydrogen	33.3	59.9	12.8
Nitrogen	126.2	227.2	33.5
Oxygen	154.8	278.6	50.1
Water	647.3	1165.1	218.2
Dichlorodifluoromethane	384.7	692.5	39.6
Dichlorofluoromethane	451.7	813.1	51.0
Ethane	305.5	549.9	48.2
Ethanol	516.3	929.3	63
Methanol	513.2	923.8	78.5
n-Butane	425.2	765.4	37.5
n-Hexane	507.9	914.2	29.9
n-Pentane	469.5	845.1	33.3
n-Octane	568.6	1023.5	24.6
Trichlorofluoromethane	471.2	848.1	43.2

Remarks.

- P, V, n, and T must have units compatible with R.
- At low temperatures or high pressures, the ideal gas law does not represent the behavior of real gases.
- No equation of state is valid for all substances over an infinite range of conditions. The Redlich-Kwong equation gives moderate to good accuracy for a variety of substances over a wide range of conditions. Results should be used with caution and tempered by experience.

Program Instructions.

1. Key the "IDEAL" program (listed on page 70) or the "RK" program (listed on page 71) into your calculator.
2. Press ■[SOLVER] IDEAL (for an ideal gas) or ■[SOLVER] RK (for Redlich-Kwong).
3. Use the variable menu to store each of the known values: key in the value and then press the corresponding menu key. Notice that there are seven menu variables. Therefore, you must use [▲] or [▼] to enter or solve for P_c.
4. Solve for the unknown by pressing the corresponding menu key.

Example 1: An Ideal Gas. What is the pressure (in bars) when 0.63 g moles of air is enclosed in a 25,000 cm^3 space at 1200K? Assume an ideal gas. From the table on page 67, R is given as 83.14. (Key the "IDEAL" program into your calculator.)

■[SOLVER] IDEAL	x: 0.0000 P V N R T
25000 V	V=25,000.0000 P V N R T
.63 N	n=0.6300 P V N R T

83.14 R

```
R=83.1400
 P | V | N | R | T |
```

1200 T

```
T=1,200.0000
 P | V | N | R | T |
```

P

```
P=2.5142
 P | V | N | R | T |
```

Example 2: Using Redlich-Kwong's Equation. The specific volume of a gas in a container is 800 cm^3/g mole. The temperature will reach 400K. Given the following constants, what will the pressure be using the Redlich-Kwong relation?

T_c = 305.5K

P_c = 48.2 atm

R = 82.05 cm^3 atm/g mole K

■ [SOLVER] RK

```
x: 2.5142
 P | V | N | R | T | TC
```

800 V

```
V=800.0000
 P | V | N | R | T | TC
```

1 N

```
n=1.0000
 P | V | N | R | T | TC
```

82.05 R

```
R=82.0500
 P | V | N | R | T | TC
```

Keys:	Display:
400 T	T=400.0000 P V N R T TC
305.5 TC	Tc=305.5000 P V N R T TC
[▼] 48.2 PC	Pc=48.2000 PC
[▲] P	P=36.2669 P V N R T TC

"IDEAL" Program Listing.

```
00 { 40-Byte Prgm }
01 LBL "IDEAL"
02 MVAR "P"
03 MVAR "V"
04 MVAR "n"
05 MVAR "R"
06 MVAR "T"
07 RCL "n"
08 RCL× "R"
09 RCL× "T"
10 RCL "P"
11 RCL× "V"
12 -
13 END
```

"RK" Program Listing.

```
00 { 113-Byte Prgm }
01 LBL "RK"
02 MVAR "P"
03 MVAR "V"
04 MVAR "n"
05 MVAR "R"
06 MVAR "T"
07 MVAR "Tc"
08 MVAR "Pc"
09 RCL "T"
10 XEQ 00
11 RCL "V"
12 XEQ 01
13 -
14 ÷
15 XEQ 01
16 4.934
17 ×
18 RCL "Tc"
19 1.5
20 Y↑X
21 XEQ 00
22 ×
23 XEQ 01
24 RCL "V"
25 +
26 LASTX
27 ×
28 RCL "T"
29 SQRT
30 ×
31 ÷
32 -
33 RCL- "P"
34 RTN
35 LBL 00
36 RCL× "n"
37 RCL× "R"
38 RTN
39 LBL 01
40 0.0867
41 RCL× "Tc"
42 XEQ 00
43 RCL÷ "Pc"
44 END
```

Black Body Thermal Radiation

All bodies emit thermal radiation according to their temperature. The higher the temperature, the more thermal radiation emitted. A black body is one that emits the maximum possible amount of energy at every wavelength for a specified temperature. The figure below represents the black body thermal emission as a function of wavelength.

This program can be used to calculate the following:

- The wavelength of maximum emissive power for a given temperature.
- The temperature corresponding to a particular wavelength of maximum emissive power.
- The total emissive power for all wavelengths at a particular temperature.
- The emissive power at a particular wavelength and temperature.



$$\lambda_{max} T = c_3$$

$$E_{b(0-x)} = \sigma T^4$$

$$E_{b\lambda} = \frac{2\pi c_1}{\lambda^5 (e^{c_2/\lambda T} - 1)}$$

Variables Used.

In Equations	Description	In Program
λ_{max}	Wavelength of maximum emissivity in microns.	W
T	Absolute temperature in °R or K.	T
$E_{b(0-x)}$	Total emissive power in Btu/hr-ft² or watts/m².	E
$E_{b\lambda}$	Emissive power at λ in Btu/hr-ft²-μm or watts/m²-μm.	P

Remarks.

- This program uses flag 00 to determine which set of units to use for the following constants:

	SI Units (Flag 00 Clear)	English Units (Flag 00 Set)
c_1	5.9544×10^{-8} W-m²	1.8887982×10^{7} Btu-μm⁴/hr-ft²
c_2	1.4388×10^{4} μm-K	2.58984×10^{4} μm-°R
c_3	2.8978×10^{3} μm-K	5.216×10^{3} μm-°R
σ	5.6697×10^{-8} W/m²-K⁴	1.713×10^{-9} Btu/hr-ft²-°R⁴

- This program uses flags 21 and 55 to control printer output.

Program Instructions.

1. Key the "BBODY" program (listed on page 76) into your calculator.
2. Clear flag 00 to select SI units (■[FLAGS] CF 00) or set flag 00 to select English units (■[FLAGS] SF 00).
3. Press [XEQ] BBODY to execute the program.
4. Use the variable menu to store each of the known values (key in the value and then press the corresponding menu key). Then calculate the desired unknown by pressing its menu key. Be sure to use values with consistent units.

To Calculate	You Must Provide
W	*T*
T	*W*
E	*T*
P	*W* and *T*

Example. If sunlight has a maximum wavelength of 0.550 μm, what is the sun's temperature in K? Assume the sun is a black body. What is the total emissive power and the emissive power at λ_{max}? What is the emissive power at λ = 0.400 μm (ultraviolet limit) and 0.700 μm (infrared limit)? (Clear flag 00 for SI units.)

■[FLAGS] CF 00
[XEQ] BBODY

```
x: 0.0000
 W | T | E | P |  |
```

.55 W

```
W=0.5500
 W | T | E | P |  |
```

T

```
T=5,268.7273
 W  T  E  P
```

The sun's temperature is 5,269K.

E

```
E=43,690,091.4860
 W  T  E  P
```

The total emissive power is approximately 43,690,000 W/m^2.

P

```
P=5.2229E-8
 W  T  E  P
```

The power at 0.550 μm is 5.2229×10^{-8} W/m^2-μm.

.4 W

```
W=0.4000
 W  T  E  P
```

P

```
P=3.9649E-8
 W  T  E  P
```

The power at 0.400 μm is 3.9649×10^{-8} W/m^2-μm.

.7 W

```
W=0.7000
 W  T  E  P
```

P

```
P=4.5934E-8
 W  T  E  P
```

The power at 0.700 μm is 4.5934×10^{-8} W/m^2-μm.

"BBODY" Program Listing. This program uses a variable menu for input. When you select a variable that you want to calculate, the program uses the character number (ASCII value) of the variable name to cause the program to branch to the appropriate subroutine (lines 13 through 15). For example, when you press T to calculate *T*, the program branches to LBL 84 because the character number of "T" is 84.

Program: **Comments:**

```
00 { 202-Byte Prgm }
01 LBL "BBODY"
02 MVAR "W"
03 MVAR "T"
04 MVAR "E"
05 MVAR "P"
06 CF 21
07 FS? 55
08 SF 21
```

Declares the menu variables and sets or clears flag 21 to match flag 55.

```
09 LBL 00
10 VARMENU "BBODY"
11 CLA
12 STOP
13 ATOX
14 X≠0?
15 XEQ IND ST X
16 GTO 00
```

Displays the variable menu and stops. Pressing [R/S] redisplays the menu. If a menu key is pressed to calculate an unknown, the character code of the variable name is used to branch to the appropriate routine.

```
17 LBL 87
18 XEQ 01
19 RCL÷ "T"
20 STO "W"
21 VIEW "W"
22 RTN
```

Calculates *W*.

```
23 LBL 84
24 XEQ 01
25 RCL÷ "W"
26 STO "T"
27 VIEW "T"
28 RTN
```

Calculates *T*.

```
29 LBL 69
30 RCL "T"
31 X↑2
32 X↑2
33 FC? 00
34 5.6697E-8
```

Calculates *E*. Flag 00 is tested to determine which value to use for the constant, σ.

```
35 FS? 00
36 1.713E-9
37 ×
38 STO "E"
39 VIEW "E"
40 RTN
```

```
41 LBL 80
42 2
43 PI
44 ×
45 FC? 00
46 5.9544E-8
47 FS? 00
48 18887982
49 ×
50 FC? 00
51 14388
52 FS? 00
53 25898.4
54 RCL÷ "W"
55 RCL÷ "T"
56 E↑X
57 1
58 -
59 RCL "W"
60 5
61 Y↑X
62 ×
63 ÷
64 STO "P"
65 VIEW "P"
66 RTN
```

Calculates P. Flag 00 is tested to enter the appropriate values for the constants c_1 and c_2.

```
67 LBL 01
68 FC? 00
69 2897.8
70 FS? 00
71 5216
72 END
```

Tests flag 00 and enters the appropriate value for the constant c_3.

5

Fluid Dynamics

Conduit Flow

This program solves for either the average velocity or the pressure drop for viscous, incompressible flow in conduits.

$$V^2 = \frac{\Delta P / \rho}{2\left(f\frac{L}{D} + \frac{K_T}{4}\right)}$$

For laminar flow (Re $\leq$ 2300)

$$f = 16/\text{Re}$$

For turbulent flow (Re $>$2300)

$$f = \frac{0.0772}{\left\{\log\left[\frac{6.9}{\text{Re}} + \left(\frac{\varepsilon}{3.7D}\right)^{1.111}\right]\right\}^2}$$

Note

Results calculated when 2300 < Re < 4000 are meaningless. To determine if results are valid, examine the calculated Reynolds number after the calculation (step 5 in the program instructions).

Variables Used.

In Equations	Description	In Program
V	Average velocity.	V
ΔP	Pressure drop.	P
L	Conduit length.	L
D	Conduit diameter. If the conduit is *not* circular, use an *equivalent* diameter defined by: $D_{eq} = 4 \times \frac{\text{Cross Sectional Area}}{\text{Wetted Perimeter}}$	D
ε	Surface irregularity.	E
Re	Reynolds number; $\text{Re} = DV/\nu$.	R
ν	Fluid kinematic viscosity.	B
ρ	Fluid density.	S
f	Fanning friction factor.*	F
K_T	Sum of fitting coefficients.	K

*The Fanning friction factor is one-forth the Darcy friction factor. That is, $f = f_D \div 4$.

Fitting Coefficients

Fitting	K
Globe valve, wide open	7.5 to 10
Angle valve, wide open	3.8
Gate valve, wide open	0.15 to 0.19
Gate valve, $^3/_4$ open	0.85
Gate valve, $^1/_2$ open	4.4
Gate valve, $^1/_4$ open	20
90° elbow	0.4 to 0.9
Standard 45° elbow	0.35 to 0.42
Tee, through side outlet	1.5
Tee, straight through	0.4
180° bend	1.6
Entrance to circular pipe	0.25 to 0.50
Sudden expansion	$(1 - A_{up}/A_{dn})^{2}$*
Acceleration from $V = 0$ to $V = V_{entrance}$	1.0

*A_{up} is the upstream area and A_{dn} is the downstream area.

Surface Irregularities

Material	ε (Feet)	ε (Meters)
Drawn or smooth tubing	5.0×10^{-6}	1.5×10^{-6}
Commercial steel or wrought iron	1.5×10^{-4}	4.6×10^{-5}
Asphalted cast iron	4.0×10^{-4}	1.2×10^{-4}
Galvanized iron	5.0×10^{-4}	1.5×10^{-4}
Cast iron	8.3×10^{-4}	2.5×10^{-4}
Wood stave	6.0×10^{-4} to 3.0×10^{-3}	1.8×10^{-4} to 9.1×10^{-4}
Concrete	1.0×10^{-3} to 1.0×10^{-2}	3.0×10^{-4} to 3.0×10^{-3}
Riveted steel	3.0×10^{-3} to 3.0×10^{-2}	9.1×10^{-4} to 9.1×10^{-3}

Program Instructions.

1. Key the "FLOW" program (listed on page 84) into your calculator.
2. Press ■[SOLVER] FLOW to select the program.
3. Use the variable menu to store the known values. Key in each value and then press the corresponding menu key. Since there are more than six variables, use the [▼] and [▲] keys to display the alternate rows of the variable menu.
4. Solve for the unknown value by pressing the corresponding menu key.
5. View the Reynolds number by pressing ■[PGM.FCN] VIEW [ENTER] R [ENTER]. If the Reynolds number is greater than 2300 and less than 4000, the results calculated are meaningless.
6. Optional: To view the Fanning friction factor, press ■[PGM.FCN] VIEW [ENTER] F [ENTER].

Example. A 60-meter pipe has three 180° bends ($K_T = 3 \times 1.6$). The fluid is water ($\nu = 9.3 \times 10^{-7}$ m²/s, $\rho = 1000$ kg/m³). The pipe diameter is 0.030 m and the surface roughness is 3×10^{-4} m. If the average velocity is 3.20 m/s, what is the pressure drop in Pascals? What is the Reynolds number? What is the Fanning friction factor?

Keys	Display
■ [SOLVER] FLOW	x: 0.0000 E D V B L K
3 [E] 4 [+/-] E	E=0.0003 E D V B L K
.03 D	D=0.0300 E D V B L K
3.2 V	V=3.2000 E D V B L K
9.3 [E] 7 [+/-] B	B=9.3000E-7 E D V B L K
60 L	L=60.0000 E D V B L K
3 [ENTER] 1.6 [×] K	K=4.8000 E D V B L K
[E] 3 [▼] S	S=1,000.0000 S P

P

```
P=418,351.2590
 S  P
```

Check the Reynolds number and the Fanning friction factor.

■[PGM.FCN] VIEW [ENTER] R [ENTER]

```
R=103,225.8065
 S  P
```

■[PGM.FCN] VIEW [ENTER] F [ENTER]

```
F=0.0096
 S  P
```

"FLOW" Program Listing.

```
00 { 128-Byte Prgm }
01 LBL "FLOW"
02 MVAR "E"
03 MVAR "D"
04 MVAR "V"
05 MVAR "B"
06 MVAR "L"
07 MVAR "K"
08 MVAR "S"
09 MVAR "P"
10 RCL "E"
11 RCL÷ "D"
12 3.7
13 ÷
14 1.111
15 Y↑X
16 RCL "V"
17 RCL× "D"
18 RCL÷ "B"
19 STO "R"
20 2300
21 X>Y?
22 GTO 01
23 R↓
24 1/X
25 6.9
26 ×
27 +
28 LOG
29 3.6
30 ×
31 1/X
32 X↑2
33 LBL 00
34 STO "F"
35 RCL× "L"
36 RCL÷ "D"
37 RCL "K"
38 4
39 ÷
40 +
41 2
42 ×
43 RCL "V"
44 X↑2
45 ×
46 RCL "P"
47 RCL÷ "S"
48 -
49 RTN
50 LBL 01
51 16
52 RCL÷ "R"
53 GTO 00
54 END
```

Flow With a Free Surface

This Solver program allows you to solve for any of the five variables in the Manning flow formula, provided the other four are known.

$$S = \frac{(nQ)^2}{2.2082\, r^{4/3} \times a^2}$$

Variables Used.

In Equation	Description	In Program
S	Slope of the bottom (dimensionless).	S
n	Roughness coefficient.	n
r	Hydraulic radius (in feet).	r
Q	Discharge flow rate (in ft³/sec).	Q
a	Cross-sectional area (in ft²).	a

Program Instructions.

1. Key the "MANN" program (listed on page 87) into your calculator.
2. Press ■[SOLVER] MANN to select the Manning flow formula Solver program.
3. Use the variable menu to store the known values. Key in each value and then press the corresponding menu key.
4. Solve for the unknown value by pressing the corresponding menu key.

Example. Find Q for $S = 0.001$, $n = 0.013$, $R = 5/12$ ft, and $A = 5.0$ ft^2.

■[SOLVER] MANN

```
x: 0.0000
 S | N | R | Q | A |
```

Keys:	Display:
.001 S	S=0.0010 S N R Q A
.013 N	n=0.0130 S N R Q A
5 [ENTER] 12 [÷] R	r=0.4167 S N R Q A
5 A	a=5.0000 S N R Q A
Q	Q=10.0826 S N R Q A

“MANN” Program Listing.

```
00 { 57-Byte Prgm }
01 LBL "MANN"
02 MVAR "S"
03 MVAR "n"
04 MVAR "r"
05 MVAR "Q"
06 MVAR "a"
07 RCL "n"
08 RCL× "Q"
09 X↑2
10 2.2082
11 ÷
12 RCL "r"
13 4
14 ENTER
15 3
16 ÷
17 Y↑X
18 ÷
19 RCL "a"
20 X↑2
21 ÷
22 RCL- "S"
23 END
```

6

Unit Conversions

Introduction

This chapter contains three programs for converting values from one system of units to another: "LENGTH," "FORCE," and "TEMP." Instructions for writing your own unit conversion programs and the conversion factors for 120 units of measure are also included.

Variable Used. *UNIT* contains the current value expressed in the base units.

Remarks.

- The following "Program Instructions" can be used with any of the programs in this chapter.
- The program tests flag 22 to determine when a number has been keyed in.
- Flags 21 and 55 are used to control printer output.

Program Instructions.

1. Key the appropriate program into your calculator. There are three sample programs listed in this chapter: "LENGTH" (page 91), "FORCE" (page 93), and "TEMP" (page 95).
2. Press [XEQ] followed by the menu key corresponding to the program (such as [XEQ] FORCE).
3. Key in the value you want to convert and then press the menu key corresponding to the *old* units.
4. Press the menu key corresponding to the *new* units. The new value is displayed. Repeat this step for as many different units as you wish.
5. To perform another conversion, go to step 3. Press [EXIT] to quit.

Example 1: Length Conversions. How many inches are in one kilometer? (Key the "LENGTH" program into your calculator.)

Keys	Display
[XEQ] LENG	x: 0.0000 CM M KM IN FT MI

Keys	Display
1 KM	1.0000 km CM M KM IN FT MI
IN	39,370.0787 in CM M KM IN FT MI

What is the same distance in feet?

Keys	Display
FT	3,280.8399 ft CM M KM IN FT MI

Now convert six miles to meters.

Keys	Display
6 MI	6.0000 mi CM M KM IN FT MI
M	9,656.0640 m CM M KM IN FT MI
[EXIT]	

Example 2: Force Conversions. Convert a 135 pound force to Newtons. (Key the "FORCE" program into your calculator.)

Keys	Display
[XEQ] FORCE	x: 9,656.0640 GF KIP N LBF PDL
135 LBF	135.0000 lbf GF KIP N LBF PDL
N	600.5099 N GF KIP N LBF PDL
[EXIT]	

"LENGTH" Program Listing. This program performs unit conversions using centimeters, meters, kilometers, inches, feet, and miles. The base unit is the meter.

Program:	**Comments:**
00 { 183-Byte Prgm } 01 LBL "LENGTH" 02 CF 21 03 FS? 55 04 SF 21	Sets or clears flag 21 to match flag 55.
05 CLMENU 06 STO "UNIT" 07 XEQ 01 08 KEY 1 XEQ 01 09 XEQ 02 10 KEY 2 XEQ 02 11 XEQ 03 12 KEY 3 XEQ 03 13 XEQ 04 14 KEY 4 XEQ 04 15 XEQ 05 16 KEY 5 XEQ 05 17 XEQ 06 18 KEY 6 XEQ 06 19 KEY 9 GTO 09 20 RCL "UNIT" 21 SF 22	Defines the unit menu. The value in the X-register is temporarily saved in the variable *UNIT*.
22 LBL A 23 MENU 24 STOP	Displays the menu and stops.
25 FC?C 22 26 GTO 00 27 × 28 STO "UNIT" 29 R↓ 30 LASTX	If flag 22 is not set (no data entry), converts the value into new units.

```
31 LBL 00
32 1/X
33 RCL× "UNIT"
34 ASTO ST L
35 CLA
36 ARCL ST X
37 ├" "
38 ARCL ST L
39 AVIEW
40 GTO A
```

Displays the new value and the unit string.

```
41 LBL 01
42 "cm"
43 0.01
44 RTN
45 LBL 02
46 "m"
47 1
48 RTN
49 LBL 03
50 "km"
51 1000
52 RTN
53 LBL 04
54 "in"
55 0.0254
56 RTN
57 LBL 05
58 "ft"
59 0.3048
60 RTN
61 LBL 06
62 "mi"
63 1609.344
64 RTN
```

Subroutines for each menu key. Each routine enters a string into the Alpha register and a conversion factor into the X-register.

```
65 LBL 09
66 END
```

Ends the program (when [EXIT] is pressed).

"FORCE" Program Listing. This program performs unit conversions using gram-force, kilopound-force, Newtons, pound-force, and poundals. The base unit is the Newton.

Program:	Comments:
`{ 193-Byte Prgm }`	
`LBL "FORCE"` `CF 21` `FS? 55` `SF 21`	Sets or clears flag 21 to match flag 55.
`CLMENU` `STO "UNIT"` `XEQ 01` `KEY 1 XEQ 01` `XEQ 02` `KEY 2 XEQ 02` `XEQ 03` `KEY 3 XEQ 03` `XEQ 04` `KEY 4 XEQ 04` `XEQ 05` `KEY 5 XEQ 05` `KEY 9 GTO 09` `RCL "UNIT"` `SF 22`	Defines the unit menu. The value in the X-register is temporarily saved in the variable *UNIT*.
`LBL A` `MENU` `STOP`	Displays the menu and stops.
`FC?C 22` `GTO 00` `×` `STO "UNIT"` `R↓` `LASTX`	If flag 22 is not set (no data entry), converts the value into new units.
`LBL 00` `1/X`	Displays the new value and the unit string.

```
31 RCL× "UNIT"
32 ASTO ST L
33 CLA
34 ARCL ST X
35 ⊢" "
36 ARCL ST L
37 AVIEW
38 GTO A
```

```
39 LBL 01
40 "gf"
41 9.80665E-3
42 RTN
43 LBL 02
44 "kip"
45 4448.22161526
46 RTN
47 LBL 03
48 "N"
49 1
50 RTN
51 LBL 04
52 "lbf"
53 4.44822161526
54 RTN
55 LBL 05
56 "pdl"
57 0.138254954380
58 RTN
```

Subroutines for each menu key. Each routine enters a string into the Alpha register and a conversion factor into the X-register.

```
59 LBL 09
60 END
```

Ends the program (when [EXIT] is pressed).

"TEMP" Program Listing. Conversions between the four temperature scales (Kelvin, °Celcius, °Fahrenheit, and °Rankine) involve additive constants as well as multiplicative factors. Without these additive constants (on lines 47, 52, 57, and 62), only relative conversions would be possible. Therefore, the "TEMP" program takes on a slightly different form than the programs listed earlier.

Note To key in lines 46 and 47 of the following program, press 1 [ENTER] [←] 0. Use the same technique (pressing [ENTER] and [←] after the first number and before the second) to key in lines 51 and 52, 56 and 57, and 61 and 62.

Program:	**Comments:**
00 { 196-Byte Prgm }	
01 LBL "TEMP" 02 CF 21 03 FS? 55 04 SF 21	Sets or clears flag 21 to match flag 55.
05 CLMENU 06 STO "UNIT" 07 XEQ 01 08 KEY 1 XEQ 01 09 XEQ 02 10 KEY 2 XEQ 02 11 XEQ 03 12 KEY 3 XEQ 03 13 XEQ 04 14 KEY 4 XEQ 04 15 KEY 9 GTO 09 16 RCL "UNIT" 17 SF 22	Defines the unit menu. The value in the X-register is temporarily saved in the variable *UNIT*.
18 LBL A 19 MENU 20 STOP	Displays the menu and stops.
21 STO "+" 22 R↓ 23 FC?C 22 24 GTO 00 25 × 26 STO "UNIT"	If flag 22 is not set (no data entry), converts the value into new units.

```
27 R↑
28 STO+ "UNIT"
29 R↓
30 R↓
31 LASTX
```

```
32 LBL 00
33 1/X
34 RCL "UNIT"
35 RCL- "+"
36 ×
37 ASTO ST L
38 CLA
39 ARCL ST X
40 ⊢" "
41 ARCL ST L
42 AVIEW
43 GTO A
```

Displays the new value and the unit string.

```
44 LBL 01
45 "°K"
46 1
47 0
48 RTN
49 LBL 02
50 "°C"
51 1
52 273.15
53 RTN
54 LBL 03
55 "°R"
56 0.5555555556
57 0
58 RTN
59 LBL 04
60 "°F"
61 0.5555555556
62 255.372222222
63 RTN
```

Subroutines for each menu key. Each routine enters a string into the Alpha register, a conversion factor into the Y-register, and an additive constant into the X-register.

```
64 LBL 09
65 END
```

Ends the program (when [EXIT] is pressed).

Writing Your Own Unit Conversion Programs

If the sample programs in this chapter do not contain the units that you work with regularly, you may want to write your own unit conversion programs.

1. Using the table at the end of this chapter (or from other reference material), select up to six units of measure.
2. Using the conversion program that follows, fill in the blank lines as follows:
   a. Line 01 should be a global label that indicates the type of units included (such as `LBL "MASS"` or `LBL "TIME"`).
   b. The two lines immediately following each label, `LBL 01` (line 41) through `LBL 06` (line 61), should include the *unit name* and the *conversion factor*.

      Each conversion factor you use should be expressed in relation to the same *base unit*. (For example, in the "LENGTH" program above, the base unit is the meter.)
   c. Optional: If you assign your unit conversion programs to the custom menu, you can make them appear to be submenus of CUSTOM by inserting `SF 27` after `LBL 09`.

The Blank Conversion Program.

```
00 { nnn-Byte Prgm }
01 LBL "program name here"
02 CF 21
03 FS? 55
04 SF 21
05 CLMENU
06 STO "UNIT"
07 XEQ 01
08 KEY 1 XEQ 01
09 XEQ 02
10 KEY 2 XEQ 02
11 XEQ 03
12 KEY 3 XEQ 03
13 XEQ 04
14 KEY 4 XEQ 04
15 XEQ 05
16 KEY 5 XEQ 05
17 XEQ 06
18 KEY 6 XEQ 06
19 KEY 9 GTO 09
20 RCL "UNIT"
21 SF 22
22 LBL A
23 MENU
24 STOP
25 FC?C 22
26 GTO 00
27 ×
28 STO "UNIT"
29 R↓
30 LASTX
31 LBL 00
32 1/X
33 RCL× "UNIT"
34 ASTO ST L
35 CLA
36 ARCL ST X
37 ├" "
38 ARCL ST L
39 AVIEW
40 GTO A
41 LBL 01
42 "unit name here"
43 conversion factor here
44 RTN
45 LBL 02
46 "unit name here"
47 conversion factor here
48 RTN
49 LBL 03
50 "unit name here"
51 conversion factor here
52 RTN
53 LBL 04
54 "unit name here"
55 conversion factor here
56 RTN
57 LBL 05
58 "unit name here"
59 conversion factor here
60 RTN
61 LBL 06
62 "unit name here"
63 conversion factor here
64 RTN
65 LBL 09
66 END
```

Dimensionless Units of Angle

Plane and solid angles are called *dimensionless* because they involve no physical dimensions. The following table relates these dimensionless units to their actual values.

Dimensionless Unit	Abbreviation	Value
Arcmin	arcmin	$\frac{1}{21600}$ unit circle
Arcsec	arcs	$\frac{1}{1296000}$ unit circle
Degree	°	$\frac{1}{360}$ unit circle
Grade	grad	$\frac{1}{400}$ unit circle
Radian	r	$\frac{1}{2\pi}$ unit circle
Steradian	sr	$\frac{1}{4\pi}$ unit sphere

Conversion Factors

The conversion factors in the following table are based on the International System of Units (SI).

If you use a unit that is not in the table, be sure that the conversion factor correctly relates the unit to the base unit for your program.

Unit	Full Name	Description	Value
a	Are	Area	100 m^2
A	Ampere	Electric current	1 A
acre	Acre	Area	4046.87260987 m^2
arcmin	Minute of arc	Plane angle	4.62962962963E−5
arcs	Second of arc	Plane angle	7.71604938272E−7
atm	Atmosphere	Pressure	101325 N/m^2
au	Astronomical unit	Length	149597900000 m
A°	Angstrom	Length	0.0000000001 m
b	Barn	Area	1.E−28 m^2
bar	Bar	Pressure	100000 N/m^2
bbl	Barrel, oil	Volume	0.158987294928 m^3
Bq	Becquerel	Activity	1 1/s
Btu	International Table Btu	Energy	1055.05585262 $Kg\text{-}m^2/s^2$
bu	Bushel	Volume	0.03523907 m^3
c	Speed of light	Velocity	299792458 m/s
C	Coulomb	Electric charge	1 A-s
cal	International Table calorie	Energy	4.1868 $Kg\text{-}m^2/s^2$
cd	Candela	Luminous intensity	1 cd
chain	Chain	Length	20.1168402337 m

Unit	Full Name	Description	Value
Ci	Curie	Activity	3.7E10 1/s
ct	Carat	Mass	0.0002 Kg
cu	US cup	Volume	2.365882365E−4 m^3
d	Day	Time	86400 s
dyn	Dyne	Force	0.00001 Kg-m/s^2
erg	Erg	Energy	0.0000001 Kg-m^2/s^2
eV	Electron volt	Energy	1.60219E−19 Kg-m^2/s^2
F	Farad	Capacitance	1 A^2-s^4/Kg-m^2
fath	Fathom	Length	1.82880365761 m
fbm	Board foot	Volume	0.002359737216 m^3
fc	Footcandle	Luminance	0.856564774909
Fdy	Faraday	Electric charge	96487 A-s
fermi	Fermi	Length	1.E−15 m
flam	Footlambert	Luminance	3.42625909964 cd/m^2
ft	International foot	Length	0.3048 m
ftUS	Survey foot	Length	0.304800609601 m
g	Gram	Mass	0.001 Kg
ga	Standard freefall	Acceleration	9.80665 m/s^2
gal	US gallon	Volume	0.003785411784 m^3
galC	Canadian gallon	Volume	0.00454609 m^3
galUK	UK gallon	Volume	0.004546092 m^3
gf	Gram-force	Force	0.00980665 Kg-m/s^2
grad	Grade	Plane angle	0.0025
grain	Grain	Mass	0.00006479891 Kg
Gy	Gray	Absorbed dose	1 m^2/s^2
h	Hour	Time	3600 s
H	Henry	Inductance	1 Kg-m^2/A^2-s^2

Unit	Full Name	Description	Value
hp	Horsepower	Power	745.699871582 $Kg\text{-}m^2/s^3$
Hz	Hertz	Frequency	1 1/s
in	Inch	Length	0.0254 m
inHg	Inches of mercury	Pressure	3386.38815789 N/m^2
inH2O	Inches of water	Pressure	248.84 N/m^2
J	Joule	Energy	1 $Kg\text{-}m^2/s^2$
kip	Kilopound-force	Force	4448.22161526 $Kg\text{-}m/s^2$
knot	Knot	Speed	0.514444444444 m/s
kph	Kilometer per hour	Speed	0.277777777778 m/s
l	Liter	Volume	0.001 m^3
lam	Lambert	Luminance	3183.09886184 cd/m^2
lb	Avoirdupois pound	Mass	0.45359237 Kg
lbf	Pound-force	Force	4.44822161526 $Kg\text{-}m/s^2$
lbt	Troy lb	Mass	0.3732417 Kg
lm	Lumen	Luminance flux	7.95774715459E−2 cd
lx	Lux	Illuminance	7.95774715459E−2 cd/m^2
lyr	Light year	Length	9.46052840488E15 m
m	Meter	Length	1 m
mho	Mho	Electric conductance	1 $A^2\text{-}s^3/Kg\text{-}m^2$
mi	International mile	Length	1609.344 m
mil	Mil	Length	0.0000254 m
min	Minute	Time	60 s
miUS	US statute mile	Length	1609.34721869 m
mmHg	Millimeter of mercury	Pressure	133.322368421 N/m^2
mol	Mole	Amount of substance	1 mol
mph	Miles per hour	Speed	0.44704 m/s

Unit	Full Name	Description	Value
N	Newton	Force	1 $Kg\text{-}m/s^2$
nmi	Nautical mile	Length	1852 m
ohm	Ohm	Electric resistance	1 $Kg\text{-}m^2/A^2\text{-}s^3$
oz	Ounce	Mass	0.028349523125 Kg
ozfl	US fluid oz	Volume	2.95735295625E−5 m^3
ozt	Troy oz	Mass	0.031103475 Kg
ozUK	UK fluid oz	Volume	0.000028413075 m^3
P	Poise	Dynamic viscosity	0.1 N/m
Pa	Pascal	Pressure	1 N/m^2
pc	Parsec	Length	3.08567818585E16 m
pdl	Poundal	Force	0.138254954376 $Kg\text{-}m/s^2$
ph	Phot	Luminance	795.774715459 cd/m^2
pk	Peck	Volume	0.0088097675 m^3
psi	Pounds per square inch	Pressure	6894.75729317 N/m^2
pt	Pint	Volume	0.000473176473 m^3
qt	Quart	Volume	0.000946352946 m^3
r	Radian	Plane angle	0.159154943092
R	Roentgen	Radiation exposure	0.000258 A-s/Kg
rad	Rad	Absorbed dose	0.01 m^2/s^2
rd	Rod	Length	5.02921005842 m
rem	Rem	Dose equivalent	0.01 m^2/s^2
s	Second	Time	1 s
S	Siemens	Electric conductance	1 $A^2\text{-}s^3/Kg\text{-}m^2$
sb	Stilb	Luminance	10000 cd/m^2
slug	Slug	Mass	14.5939029372 Kg
sr	Steradian	Solid angle	7.95774715459E−2

Unit	Full Name	Description	Value
st	Stere	Volume	1 m^3
St	Stokes	Kinematic viscosity	0.0001 m^2/s
Sv	Sievert	Dose equivalent	1 m^2/s^2
t	Metric ton	Mass	1000 Kg
T	Tesla	Magnetic flux	1 $Kg/A\text{-}s^2$
tbsp	Tablespoon	Volume	1.47867647813E−5 m^3
therm	EEC therm	Energy	105506000 $Kg\text{-}m^2/s^2$
ton	Short ton	Mass	907.18474 Kg
tonUK	Long ton	Mass	1016.0469088 Kg
torr	Torr	Pressure	133.322368421 N/m^2
tsp	Teaspoon	Volume	4.92892159375E−6 m^3
u	Unified atomic mass	Mass	1.66057E−27 Kg
V	Volt	Electric potential	1 $Kg\text{-}m^2/A\text{-}s^3$
W	Watt	Power	1 $Kg\text{-}m^2/s^3$
Wb	Weber	Magnetic flux	1 $Kg\text{-}m^2/A\text{-}s^2$
yd	International yard	Length	0.9144 m
yr	Year	Time	31556925.9747 s
°	Degree	Angle	2.77777777778E−3
°C	Degree Celsius	Temperature	1 K*
°F	Degree Fahrenheit	Temperature	0.555555555556 K*
K	Degree Kelvin	Temperature	1 K*
°R	Degree Rankine	Temperature	0.555555555556 K*
μ	Micron	Length	0.000001 m

*Refer to the "TEMP" program listing starting on page 94.

Step-by-Step Solutions for Your HP-42S Calculator

Mechanical Engineering contains a variety of programs and examples to provide solutions for mechanical engineers and engineering students.

- **Forces and Vectors**
  Vectors in the HP-42S • Static Equilibrium at a Point
- **Equations of Motion**
- **Analysis Programs**
  Mohr Circle Analysis • Soderberg's Equation for Fatigue • Composite Section Properties • Spring Constant • Linear or Angular Deformation
- **Thermodynamics**
  Equations of State • Black Body Thermal Radiation
- **Fluid Dynamics**
  Conduit Flow • Flow With a Free Surface
- **Unit Conversions**
  Writing Your Own Unit Conversion Programs • Dimensionless Units of Angle • Conversion Factors

HEWLETT PACKARD

Reorder Number
00042-90022

00042-90024 English
Printed in Canada 10/88

0 88698 00038 0